

NAG Fortran Library Routine Document

F07GWF (CPPTRI/ZPPTRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07GWF (CPPTRI/ZPPTRI) computes the inverse of a complex Hermitian positive-definite matrix A , where A has been factorized by F07GRF (CPPTRF/ZPPTRF), using packed storage.

2 Specification

```

SUBROUTINE F07GWF(UPLO, N, AP, INFO)
ENTRY      cpptri (UPLO, N, AP, INFO)
INTEGER   N, INFO
complex AP(*)
CHARACTER*1 UPLO

```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To compute the inverse of a complex Hermitian positive-definite matrix A , this routine must be preceded by a call to F07GRF (CPPTRF/ZPPTRF), which computes the Cholesky factorization of A using packed storage.

If UPLO = 'U', $A = U^H U$ and A^{-1} is computed by first inverting U and then forming $(U^{-1})(U^{-1})^H$.

If UPLO = 'L', $A = LL^H$ and A^{-1} is computed by first inverting L and then forming $(L^{-1})^H(L^{-1})$.

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Parameters

- 1: UPLO – CHARACTER*1 *Input*
On entry: indicates whether A has been factorized as $U^H U$ or LL^H as follows:
 if UPLO = 'U', $A = U^H U$, where U is upper triangular;
 if UPLO = 'L', $A = LL^H$, where L is lower triangular.
Constraint: UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 3: AP(*) – *complex* array *Input/Output*
Note: the dimension of the array AP must be at least $\max(1, N * (N + 1) / 2)$.

On entry: the upper triangular matrix U stored in packed form if UPLO = 'U' or the lower triangular matrix L stored in packed form if UPLO = 'L', as returned by F07GRF (CPPTRF/ZPPTRF).

On exit: U is overwritten by the upper triangle of A^{-1} if UPLO = 'U'; L is overwritten by the lower triangle of A^{-1} if UPLO = 'L'. More precisely, the (i, j) th element of A^{-1} is stored in AP($i + j(j - 1)/2$) for $i \leq j$ if UPLO = 'U', and in AP($i + (2n - j)(j - 1)/2$) for $i \geq j$ if UPLO = 'L'.

4: INFO – INTEGER

Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO = i , the i th diagonal element of the Cholesky factor is zero; the Cholesky factor is singular and the inverse of A cannot be computed.

7 Accuracy

The computed inverse X satisfies

$$\|XA - I\|_2 \leq c(n)\epsilon\kappa_2(A) \quad \text{and} \quad \|AX - I\|_2 \leq c(n)\epsilon\kappa_2(A),$$

where $c(n)$ is a modest function of n , ϵ is the *machine precision* and $\kappa_2(A)$ is the condition number of A defined by

$$\kappa_2(A) = \|A\|_2 \|A^{-1}\|_2.$$

8 Further Comments

The total number of real floating-point operations is approximately $\frac{8}{3}n^3$.

The real analogue of this routine is F07GJF (SPPTRI/DPPTRI).

9 Example

To compute the inverse of the matrix A , where

$$A = \begin{pmatrix} 3.23 + 0.00i & 1.51 - 1.92i & 1.90 + 0.84i & 0.42 + 2.50i \\ 1.51 + 1.92i & 3.58 + 0.00i & -0.23 + 1.11i & -1.18 + 1.37i \\ 1.90 - 0.84i & -0.23 - 1.11i & 4.09 + 0.00i & 2.33 - 0.14i \\ 0.42 - 2.50i & -1.18 - 1.37i & 2.33 + 0.14i & 4.29 + 0.00i \end{pmatrix}.$$

Here A is Hermitian positive-definite, stored in packed form, and must first be factorized by F07GRF (CPPTRF/ZPPTRF).

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F07GWF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX
PARAMETER       (NMAX=8)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
CHARACTER       UPLO
*      .. Local Arrays ..
complex        AP(NMAX*(NMAX+1)/2)
CHARACTER       CLABS(1), RLABS(1)
*      .. External Subroutines ..
EXTERNAL        cpptrf, cpptri, X04DDF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07GWF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
      END IF
*
*      Factorize A
*
      CALL cpptrf(UPLO,N,AP,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*      Compute inverse of A
*
        CALL cpptri(UPLO,N,AP,INFO)
*
*      Print inverse
*
        IFAIL = 0
        CALL X04DDF(UPLO,'Nonunit',N,AP,'Bracketed','F7.4',
+                'Inverse','Integer',RLABS,'Integer',CLABS,80,0,
+                IFAIL)
      ELSE
        WRITE (NOUT,*) 'A is not positive-definite'
      END IF
      END IF
      STOP
*
      END

```

9.2 Program Data

```

F07GWF Example Program Data
  4                                     :Value of N
  'L'                                   :Value of UPLO
(3.23, 0.00)
(1.51, 1.92) ( 3.58, 0.00)
(1.90,-0.84) (-0.23,-1.11) ( 4.09, 0.00)
(0.42,-2.50) (-1.18,-1.37) ( 2.33, 0.14) ( 4.29, 0.00) :End of matrix A

```

9.3 Program Results

F07GWF Example Program Results

```
Inverse
          1              2              3              4
1  ( 5.4691, 0.0000)
2  (-1.2624,-1.5491) ( 1.1024, 0.0000)
3  (-2.9746,-0.9616) ( 0.8989,-0.5672) ( 2.1589, 0.0000)
4  ( 1.1962, 2.9772) (-0.9826,-0.2566) (-1.3756,-1.4550) ( 2.2934, 0.0000)
```
